Die Programmierung der ATMEL AVR Controller mit BASCOM



A) HARDWARE



Beispiel für ein Minimalsystem mit AT90C23213:

�

Die Hardware für einen Mega8  kann ähnlich oder sogar noch einfacher ausfallen, wenn der interne RC-Oszillator benutzt wird.





Mit Quarz (und 2x 22pF):

 

��Mit internem Oszillator:



���

Achtung: 

Die Werkseinstellung der Fuse Bits ist die für den internen Oszillator!

Wenn ein Quarz benutzt wird muss Fuse Bit A987 auf 1111:1111 eingestellt werden.

Am besten dann auch die Brown out detection aktivieren (Fusebit B: BODEN enabled).

(siehe SOFTWARE)







Zum Programmieren wird der Controller mit dem PC über einen Programmieradapter verbunden. Es gibt unterschiedliche Schaltungen, z.B. die folgende zum Programmieren über den Parallelport:

�



B) SOFTWARE: BASCOM

1. BASCOM - Optionen einstellen / überprüfen�

Options - Compiler - Chip: richtiger Baustein ausgewählt?�

Options - Programmer �            -  STK200 / STK300�            - Adresse des Parallelports (LPT1 = 378h, LPT2 = 278h)



�2. Programm schreiben, compilieren und zum Controller schicken�

File - New�Quelltext schreiben�

Program - Compile�

Program - Send to chip��Bei umfangreicheren Programmen kann es nützlich sein, das Programm im eingebauten Simulator  auszutesten.









Programmstruktur (Beispiel):



‘Deklarationen + Initialisationen:

	$CRYSTAL  = 4000000	

$Regfile = „2313def.dat„		

	Config Portc = Output

	Config Portb = Input�	Config ....



	DIM VariableA as byte

	DIM VariableB as integer

	....



‘Hauptprogramm ist sehr oft eine Endlosschleife

	DO

		......

		......

	LOOP

	END









Stolpersteine:



Chip nicht korrekt eingestellt�

Vergessen, Port als Ein- oder Ausgang zu konfigurieren�

Wenn das Programm keine Endlosschleife oder keinen END-Befehl enthält, wird erst das Programm ausgeführt, danach aber sinnloser Schrott, nämlich die Befehle die sich aus den zufälligen Speicherinhalten hinter dem  letzten Befehl ergeben.�

Fusebits für Mega8 und Quarzbeschaltung einstellen:

�  Manuel Program - Lock and Fuse bits - Fuse bit A987 - 1111:1111 external crystal�

C) Erste Schritte....

Port schreiben und lesen



Konfiguration als Ein/Ausgang:�z.B.�Config PortB = Output / Input für den ganzen Port B�Config PinC.3 = Output für Bit 3 von PortC��Alternativ dazu kann man auch direkt in die Data Direction Register schreiben:�z.B.�DDRD = &B0001100�                   (1 = Ausgang, 0 = Eingang)

�

Werte setzen:�PortA = &H1A�PortC.4 = 1 �

Werte lesen �DIM x as byte�x = PinA��DIM y as bit�y = PinC.6



Details zu internem Pullup siehe „Taster„

�



Aufgabe 1



Damit fängt man immer an, wenn man Neuling ist, oder einen neuen Controller benutzt:

Schreibe ein Programm, welches eine oder mehrere LEDs zum Blinken bringt.��Überlege:

Wie sind die LEDs angeschlossen? Braucht man H oder L damit sie leuchten?

Welcher Wert muss am Port ausgegeben werden, damit z.B. die dritte LED von links leuchtet?  (binär -> hex, evtl. dez.)





Benutze die Befehle

$CRYSTAL = 

CONFIG PortX = Output

DO ... LOOP

Wait

PortX = Konstante



Alle Befehle sind in der BASCOM - Hilfe  erklärt.



'BLINK1



$crystal = 4000000

Config Portb = Output



Do

   Portb = 0

   Wait 1

   Portb = &HFF

   Wait 1

Loop





�Aufgabe 2



Sollen die LEDs etwas schneller blinken?



Versuch’s mit dem Befehl Waitms.



'BLINK2

$crystal = 4000000

Config Portb = Output



Do

   Portb = 0

   Waitms 100

   Portb = &HFF

   Waitms 100

Loop



Aufgabe 3



Wie wär’s jetzt mit einem Lauflicht?



Wir fangen mit einer leuchtenden LED an und schieben die Bits hin und her mit ROTATE, einmal links herum und einmal rechts herum. Das ganze in einer Schleife, damit es ewig so weitergeht. 



$crystal = 4000000



Config Portb = Output

Dim I As Byte

Dim X As Byte



X = 1



Do

   'links

    For I = 1 To 7

       Portb = X

       Waitms 100

       Rotate X , Left , 1

    Next I



   'rechts

    For I = 7 To 1 Step -1

       Portb = X

       Waitms 100

       Rotate X , Right , 1

    Next I



Loop

LCD-Display



Aufgabe 4: Ansteuerung eines LCD-Displays



Das hätte eigentlich schon früher kommen sollen, denn jeder Programmierkurs, der etwas auf sich hält, beginnt damit.�

Erraten, es handelt sich um das berühmt-berüchtigte „Hello world!„ - Programm.

Versuche den Text auf dem LCD - Display auszugeben. Beachte dabei die Anschlüsse des 

LCDs. Auf dem Lauth’schen Board ist die Verbindung standardmässig:

�

Zusätzlich ist RW mit Masse zu verbinden!

(Es wird mit nur 4 Datenbits gearbeitet)

(Beim AT90C2313 gibt es keinen PortC, hier muss PortB oder PortD benutzt und entsprechen konfiguriert werden.)



Im Programm muss das LCD mit 2 Befehlen konfiguriert werden:



Config LCD legt die Grösse fest (20 * 2)

Config Lcdpin legt die Anschlussbelegung fest�(kann auch geändert werden, wenn nötig)



Vor dem Schreiben soll das Display mit CLS gelöscht werden. Dies bewirkt auch einen LCD-Reset.



Das Schreiben selbst erfolgt mit dem Befehl LCD





Erweiterungen:



Mit LOCATE kann der Cursor positioniert werden, so dass man z.B. auch in die 2.Zeile schreiben kann. 



'$LCD1

Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



Cls                                                         ' CLS = Clear and Reset LCD-Display!

Cursor Blink                                                ' CURSOR blinkt!

'_____________________________________________________

Lcd "Hello world!"                                          ' erste Zeile!

Locate 2 , 1                                                ' Cursor auf 2 Zeile, erstes Zeichen!

Lcd "LCD-Test"

End

Serielle Schnittstelle



Aufgabe 5: Kontakt mit einem PC�Senden eines Textes



Es soll eine serielle Übertragung vom Mikrokontroller zum PC prohrammiert werden.



Die RS232 - Signale des PCs haben (12V, während der Controller mit 5V-Pegel arbeitet. Ausserdem ist die Logik invertiert.



Es muss also eine Anpass-Schaltung zwischengeschaltet werden, z.B. die klassische MAX232- Schaltung, welche in der hier abgebildeten Form ihren „Saft„ eleganterweise aus der RS232-Schnittstelle des PCs beziehen kann , wenn J1 gesetzt ist (Voraussetzung ist natürlich ein 1:1-Kabel, welches auch die Signale der Pins 4 (DTR) und 7 (RTS) führt).

Die Speisung kann aber auch vom Controller-Board aus erfolgen, wenn J2 gesetzt ist.

�

D1, D2 =  1N4148

D3        =  ZD4v7

R1         =  100(

C1...C5  = 1(F�J1 = Stromversorgung vom PC

J2 = Stromversorgung von der 

        Kontrollerplatine��

Die Verbindung nach rechts zum Mikrokontroller ist die folgende:



Pin 3�RxD�PortD.0  (RxD)��Pin 2�TxD�PortD.1  (TxD)��

Die Anpass-Schaltung kann übrigens auch mit 2 Transistor-Invertern aufgebaut werden. 

An RxD des PCs (Pin 2) genügt meist schon ein 5V-Pegel, so dass die Schaltung nur 5V Betriebsspannung braucht. 



��



T1 (npn) an RxD: z.B BC547 o.ä.





T2 (pnp) an TxD: z.B. BC557 o.ä. ��





Benutze für das Testprogramm die Befehle:



BAUD = ...

PRINT .....



(Am besten Text in einer Endlosschleife ausgeben mit Verzögerung, dies ist günstig wenn das Signal mit dem Oszilloskop beobachtet werden soll).



Als Terminal kann das in BASCOM eingebaute benutzt werden.



Achtung!�     Die Baudraten und sonstigen Parameter der Schnittstellen von PC und Controller �      müssen natürlich übereinstimmen!



$crystal = 4000000



Baud = 2400

Do

  Print "Hello world"

  Waitms 200

Loop



End







Aufgabe 6: Senden eines Variableninhaltes



Statt eines langweiligen konstanten Textes soll der Controller den Inhalt einer jede 200ms inkrementierten Variablen übermitteln (dezimal und hexadezimal).



Befehle:

	PRINT  X;		‘X = Zählvariable (vorher dimensionieren)

	PRINT  HEX(X)



Welchen Unterschied macht es, ob hinter dem PRINT ein „;„ angefügt ist oder nicht?



'$PRINT2.BAS



$crystal = 4000000



Baud = 2400

Dim I As Long



Print "Hello world"



Do

   I = I + 1

  Print I;

  Print "     ";

  Print Hex(i)



  Waitms 200

Loop



End





Mehrere Controller an einer RS232-Schnittstelle



�



Der TxD-Ausgang des AVR hat im Ruhezustand +5V, die L-Pegel bei der Übertragung bewirken über die Diode Basisstrom für den pnp-Transistor und H am PC-Stecker.�

Die RxD-Eingänge liegen parallel so dass alle AVRs gleichzeitig „hören“ können.�

Wichtig ist, dass keine 2 AVRs gleichzeitig senden.









�Aufgabe 7: Empfang eines Textes vom PC



Der Mikrokontroller soll den Bediener am anderen Ende der Leitung auffordern, seinen Namen einzugeben. Anschliessend soll der Mikrokontroller antworten mit „Hallo  Name„



Befehl:

	INPUT 



Beachte, dass zuerst eine Variable zum Aufnehmen des empfangenen Textes dimensioniert werden muss, z.B. mit  DIM S as STRING * 10  (max. 10 Zeichen lang)

 



'$input.BAS

$crystal = 4000000

Baud = 2400

Dim S As String * 10



Do

 	Input "Please type your name:" , S



  	Print "Hello ";

  	Print S



Loop

End









Erweiterung:

Der Controller  soll zusätzlich nach dem Alter fragen und dies ebenfalls ausgeben.�(Welcher Variablentyp eignet sich für das Alter? Informiere dich über die unterschiedlichen Variablentypen!)



Damit die Anzeige im Terminal übersichtlicher wird, soll am Ende jeweils eine Leerzeile ausgegeben werden.





'$input.BAS

$crystal = 4000000

Baud = 2400

Dim S As String * 10



Do

 	Input "Please type your name:" , S



  	Print "Hello ";

  	Print S



Loop

End









�Software-UART



Diese Option ist interessant wenn man mehrere serielle Schnittstellen braucht, der Controller hardwaremässig aber nur eine anbietet.



Die Pins für TxD und RxD müssen jeweils einzeln mit dem OPEN - Befehl als „Gerät“ Nummer x konfiguriert werden (anglehnt an die DOS-Syntax).

Anschliessend werden sie mit dem PRINT #x beschrieben oder mit INPUT #x bzw INKEY (#x) gelesen.



Beispiel für das Senden:

$baud = 9600

'open channel for output

Open "comd.3:9600,8,n,1" For Output As #1

Print #1 , "serial output"





Beispiel für Empfangen und Senden:

$baud = 9600

'open channel for output

Open "comd.3:9600,8,n,1" For Output As #1

Open "comd.2:9600,8,n,1" For Input As #2

Print #1 , "serial output"







Do

    I = Inkey(#2)

    If I <> 0 Then

     	 I = I + 1

      	Print #1 , I

     End If

Loop



Close #1

Close #2

End



(In den beiden Beispielprogrammen ist das CLOSE eigentlich unwirksam, wenn die Schnittstelle nicht mehr benutzt wird, sollte sie aber geschlossen werden.)



Weitere interessante Befehle:

INPUT, INPUTBIN, INPUTHEX

PRINT

ISCHARWAITING, WAITFORKEY





�

Taster und Schalter an den Ports



Aufgabe 8: Eingabe über Taster



Ein Portpin soll abgefragt und das Ergebnis mit den LEDs eines anderen Ports angezeigt werden.



Achtung: ein Eingang wird mit PinX.Y, nicht mit PortX.Y gelesen!



Achtung: bei der Funktion als Eingang kann der Pin mit oder ohne internen Pullup-Widerstand betrieben werden!





Portfunktion in Abhängigkeit von Data Direction Register (Config PortX = Input/Output) und gesetztem Wert (PortX = 1/0):



Config =

(DDRx)�PortX = �Pullup���0  Input�0�Nein�Hochohmiger Eingang��0  Input�1�Ja�Eingang mit Pullup - Widerstand��1  Output�0�Nein�Ausgang mit log. 0��1  Output�1�Nein�Ausgang mit log. 1��





Taster nach +5V:

In diesem Fall muss ein Pulldown-Widerstand von z.B. 10k( verwendet werden, da sonst der Eingang bei nicht gedrücktem Taster offen liegt. (Die 3 Taster (, (, ENTER des Lauth’schen Boardes sind schon so beschaltet).



'$READ_PIN1.BAS

Config Portb = Output

Config Pind.3 = Input



Dim Pinstate As Bit



'Offener Eingang für Einsatz mit Taster gegen +5V, mit Pulldown-Widerstand:

'Ohne Pulldown ist der Eingang hochohmig (CMOS)

Portd.3 = 0



Do

   Pinstate = Pind.3

   If Pinstate = 1 Then

      'alle LEDs an

       Portb = &HFF

   Else

       'alle LEDs aus

       Portb = 0

   End If

Loop



End





�Taster nach GND

Dies ist oft praktischer (und daher in der (C-Technik verbreitet), da man keine +5V auf die Taster führen muss, ein Masseanschluss ist sowieso immer irgendwo zu finden. Ausserdem kann ein AVR einen internen Pullup- Widerstand zur Verfügung stellen, so dass kein externer nötig ist.

��

Die gezeichnete Konfiguration des Tasters arbeitet mit einem programmierten internen Pullup-Widerstand.�

Um diesen einzuschalten muss der Portpin  als Input konfiguriert und auf log. 1 gesetzt werden:



CONFIG PinD.0 as INPUT

PortD.0 = 1 



Gelesen wird der Pin als  PinD.0��

Achtung:

Bei Eingängen mit Schalter gegen Masse:�Es dürfen keine LEDs parallel zu den Eingängen liegen (wie bei der älteren Version des Lauth’schen des Testboards), denn diese ziehen das Potential zu stark herunter (auf 2V), so dass es für H nicht mehr reicht!



'$READ_PIN2.BAS

'Pin D3 lesen und anzeigen  LEDs von PortB

'Taster gegen Masse, uC mit internem Pullup

'Es darf keine LED parallel zum Eingang liegen!!



Config Portb = Output

Config Pind.3 = Input



'_____________________________________________________

Dim Pinstate As Bit



' Port=1 zum Einschalten des internen Pullup:

Portd.3 = 1



Do

   Pinstate = Pind.3

   If Pinstate = 1 Then

      'alle LEDs an

       Portb = &HFF

   Else

       'alle LEDs aus

       Portb = 0

   End If

Loop



End





�Erweiterung:

Ein Portpin soll abgefragt und das Ergebnis mit LEDs, auf dem LCD-Display und an einen angeschlossenen PC über die RS232-Schnittstelle übermittelt werden.



'$READ_PIN3.BAS

'Pin D3 lesen und anzeigen mit LCD an Port C (C7..C4), PRINT, LEDs von PortB



'Config hardware



Config Portb = Output

Config Pind.3 = Input



Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



$baud = 2400



Cls                                     ' CLS = Clear and Reset LCD-Display!



'_____________________________________________________

Dim Pinstate As Bit



' Port=1 zum Einschalten des internen Pullup:

Portd.3 = 1



Do

   Pinstate = Pind.3

   If Pinstate = 1 Then

      'alle LEDs an

       Portb = &HFF



       Cls

       locate 1,1

       Lcd "HIGH"

       Print "HIGH"

   Else

       'alle LEDs aus

       Portb = 0



       Cls

       locate 1,1

       Lcd "LOW"

       Print "LOW"

   End If

Loop



End





�Matrixtastatur

'$Matrixtastatur_JC_Code_1.bas

'Matrixtastatur  an PortD

'PortA zeigt an



Config Porta = Output                                       'soll key anzeigen



Ddrd = &H0F                                                 'B0...B3 = Output, B4...B7 = Input

Portd = &HFF                                                'mit Pullup



Dim Col As Byte

Dim Row As Byte

Dim Key As Byte





Do

   For Col = 0 To 3

      'Spalten werden der Reihe nach auf null gelegt

      Select Case Col

         Case 0 : Reset Portd.0

         Case 1 : Reset Portd.1

         Case 2 : Reset Portd.2

         Case 3 : Reset Portd.3

      End Select



      'Abfrage der Zeilen

      For Row = 0 To 3

           Select Case Row

               Case 3 : Debounce Pind.7 , 0 , Calc_key , Sub

               Case 2 : Debounce Pind.6 , 0 , Calc_key , Sub

               Case 1 : Debounce Pind.5 , 0 , Calc_key , Sub

               Case 0 : Debounce Pind.4 , 0 , Calc_key , Sub

           End Select

      Next



      'wieder hoch setzen

      Portd = &HFF



   Next

Loop





Calc_key:

   Select Case Row

        Case 0 : Key = Col

        Case 1 : Key = Col + 4

        Case 2 : Key = Col + 8

        Case 3 : Key = Col + 16

   End Select



   Porta = Key

Return



$Matrixtastatur_JC_Code_1.bas

'Matrixtastatur  an PortD

'PortA zeigt an



Keyport Alias Portd

Keyportddr Alias Ddrd



Config Porta = Output                                       'soll key anzeigen



Keyportddr = &H0F                                           'B0...B3 = Output, B4...B7 = Input

Keyport = &HFF                                              'mit Pullup



Dim Col As Byte

Dim Row As Byte

Dim Key As Byte





Do

   For Col = 0 To 3

      'Spalten werden der Reihe nach auf null gelegt

      Keyport = Keyport And(2 ^ Col)

      'Select Case Col

       '  Case 0 : Reset Portd.0

       '  Case 1 : Reset Portd.1

       '  Case 2 : Reset Portd.2

       '  Case 3 : Reset Portd.3

      'End Select



      'Abfrage der Zeilen

      For Row = 0 To 3

           Select Case Row

               Case 3 : Debounce Pind.7 , 0 , Calc_key , Sub

               Case 2 : Debounce Pind.6 , 0 , Calc_key , Sub

               Case 1 : Debounce Pind.5 , 0 , Calc_key , Sub

               Case 0 : Debounce Pind.4 , 0 , Calc_key , Sub

           End Select

      Next



      'wieder hoch setzen

      Portd = &HFF



   Next

Loop





Calc_key:

   Select Case Row

        Case 0 : Key = Col

        Case 1 : Key = Col + 4

        Case 2 : Key = Col + 8

        Case 3 : Key = Col + 16

   End Select



   Porta = Key

Return









�Externe Interrupts



Pins für externe Interrupts:



INT0�PortD.2��INT1�PortD.3��

Konfigurationsmöglichkeiten:

Config Intx= Rising / Falling / Low Level





'$Interrupt_1.bas

'Bei jeder steigenden Flanke an INT0 (PortD.2) soll PortB.7 = 1 geschaltet werden,

'bei jeder fallenden Flanke an INT1 (PortD.3) soll PortB.7 = 0 werden



Config Pinb.7 = Output

Config Pind.2 = Input

Config Pind.3 = Input



Portd.2 = 1                             'Pullup ein

Portd.3 = 1                             'Pullup ein



On Int0 Int0serv

On Int1 Int1serv



Config Int0 = Rising

Config Int1 = Falling

Enable Int0

Enable Int1

Enable Interrupts



'-------------------

Do

   'Hauptprogramm : tu nix

Loop

'--------------------



Int0serv:

   Portb.7 = 1

Return



Int1serv:

   Portb.7 = 0

Return









'$Interrupt_2.bas

'Bei jeder steigenden Flanke an INT0 (PortD.2) soll eine Variable inkrementiert werden,

'bei jeder fallenden Flanke an INT1 (PortD.3) soll eine Variable dekrementiert werden

'Die Variable wird über LEDs an PortB angezeigt

'Achtung: bei einfachen Tastern nach Masse entprellen mit 100nF nach +5V



Dim Zaehler As Byte



Config Portb = Output

Config Pind.2 = Input

Config Pind.3 = Input



Portd.2 = 1                             'Pullup ein

Portd.3 = 1                             'Pullup ein



On Int0 Int0serv

On Int1 Int1serv



Config Int0 = Rising

Config Int1 = Falling

Enable Int0

Enable Int1

Enable Interrupts



Zaehler = 0



'-------------------

Do

   'Hauptprogramm : tu nix

Loop

'--------------------



Int0serv:

   Zaehler = Zaehler + 1

   Portb = Zaehler

Return



Int1serv:

   Zaehler = Zaehler - 1

   Portb = Zaehler

Return





Achtung:

Der Sprung in die Interrupt-Routine kostet Zeit!

Gesichert werden automatisch immer die Register R0...R31 (je 1 x PUSH und 1 x  POP), also 32 ( 2 = 64 Zyklen, d.h. 64 ( 0.25(s = 16(s bei einem 4MHz-Quarz! 



Ein Verbesserung kann Nosave bringen, wenn man sich manuell um das Sichern der benutzten Register und des Statusflags kümmert. 

�Wenn es mal schnell gehen muss: ASM und BAS mischen

siehe Artikel Assembler für Basic-Programmierer





$asm

........ASM - Code

$end asm



Achtung: ASM-Labels müssen ganz links und allein in einer Zeile stehen.



Wichtige ASM-Befehle (Beispiele):



sbi  portd, 5		Set Bit immediate: 	D.5 = 1

cli   portd.5		Clear bit immediate:  	D.5 = 0

ldi  R24, 56		Load immediate: 	R24 = 56

out  PortB, R24	(Output geht nicht direkt, nur über Register)

in R24, PortB



.def  R24 = MyReg	(ALIAS definieren)



mov r24, r25 

nop			No operation  (1 Takt warten)  



  � Timer als Timer



bei 4MHz - Quarz:



Prescaler�1�8�64�256�1024��Auflösung in (s (TCLK)�0.25�2�16�64�256��Tmax Timer0 (8 bit)�64(s�512(s�4096(s�16384(s�65 536(s��Tmax Timer1 (16 bit)�16.384ms�131.072ms�1.048s�4.194s�16.777s��

Innerhalb der Timerservice - Routine wird der Timer mit einem Preload - Wert geladen

und er zählt dann bis zum Überlauf:



T = Tmax - Preload ( TCLK

oder:

	T = (256 - Preleoad) ( TCLK		(Timer0)

	T = (65536 - Preleoad) ( TCLK		(Timer1)



Das Laden des Timers muss als erstes in der Interrupt - Service - Routine geschehen, da der Timer sofort mit dem Zählen beginnen muss!



Für Timer1 gibt es die elegante Möglichkeit des Output Compare: in diesem Fall wird der Preload-Wert automatisch hardwaremässig geladen, die Interruptroutine braucht sich nicht darum zu kümmern.



Zu berücksichtigen ist auch, dass das Sichern der Register beim Aufruf der Timer-Routine  eine nicht zu unterschätzende Zeit benötigt!

Gesichert werden die Register R0...R31 (je 1 x PUSH und 1 x  POP), also 32 ( 2 = 64 Zyklen, d.h. 64 ( 0.25(s = 16(s bei einem 4MHz-Quarz. 



Ein Verbesserung kann Nosave bringen, wenn möglich.



Noch besser wäre es, zeitkritisches in ASM zu schreiben.





'$Timer0_2.bas

'8-bit-Timer als Rechteckgenerator

'mit Voreinstellung

'Achtung: bei kleinen Zeiten ist die Zeit zur Rettung der Register nicht zu vernachlässigen!

'Nosave bringt, wenn möglich, eine starke Verbesserung



$crystal = 4000000



Config Portb = Output



On Timer0 Timer0serv                    ''gemessen:  56us statt 40us

'On Timer0 Timer0serv Nosave             '' gemessen: 42us statt 40us



Config Timer0 = Timer , Prescale = 8    'Timer-CLK = 500kHz -> TCLK = 2us

Enable Timer0

Enable Interrupts



Dim X As Bit

Dim Preload As Byte



Preload = 236		‘256-236 = 20 Takte = 20 *2us = 40us

Timer0 = Preload



'------------------

Do

  'tu nix

Loop

'------------------

Timer0serv:

'20 * 2us = 40us

    Timer0 = Preload

    X = Not X

    Portb.0 = X

Return



Welche Register werden von der Service-Routine benutzt?

ASM-Listing von Timer0serv:



avr0038:  ldi    XL, 0x61     ; 0038 E6A1	;Timer0serv

          ldi    XH, 0x00     ; 0039 E0B0

          ld     r24, X       ; 003A 918C	;r24 = Preload (Adr. 61h)



          out    TCNT0, r24   ; 003B BF82	;TCNT0 = Preload



          ldi    XL, 0x60     ; 003C E6A0

          ldi    XH, 0x00     ; 003D E0B0

          ld     r24, X       ; 003E 918C	;Toggle-Bit u. Rest (Adr. 60h)-> r24



          bst    r24, 7       ; 003F FB87	

          clr    r24          ; 0040 2788

          bld    r24, 0       ; 0041 F980

          com    r24          ; 0042 9580

          bst    r24, 0       ; 0043 FB80	;Toggle-Bit ist invertiert u. -> r24



          ldi    XL, 0x60     ; 0044 E6A0

          ldi    XH, 0x00     ; 0045 E0B0

          ld     r24, X       ; 0046 918C

          bld    r24, 7       ; 0047 F987

          st     X, r24       ; 0048 938C		



          ldi    XL, 0x60     ; 0049 E6A0

          ldi    XH, 0x00     ; 004A E0B0

          ld     r24, X       ; 004B 918C	



          bst    r24, 7       ; 004C FB87	



          clr    XH           ; 004D 27BB

          ldi    XL, 0x38     ; 004E E3A8

          ld     r24, X       ; 004F 918C

          bld    r24, 0       ; 0050 F980

          st     X, r24       ; 0051 938C	;

          reti                ; 0052 9518	;Ende Timer0serv.



Nur R24 wird als Zwischenspeicher benutzt, und das Tflag wird verändert, also sollte eigentlich das Statusregister gesichert werden!







Realisierung in ASM mit nur 3.5(s Zeitverlust (2(s ohne Sichern des Statusregisters):



'$Timer0_3.bas

'8-bit-Timer als Rechteckgenerator

'Interrupt-Routine in ASM: '' gemessen: 7.5us statt 4us



$crystal = 4000000

$regfile = "8535def.dat"





Config Portb = Output



On Timer0 Timer0serv Nosave



Config Timer0 = Timer , Prescale = 8    'Timer-CLK = 500kHz

Enable Timer0

Enable Interrupts



Dim X As Bit

Dim Preload As Byte



Preload = 254                           '256-254=2 Zyklen

Timer0 = Preload



'------------------

Do

  'tu nix

Loop

'------------------

Timer0serv:

'2* 2us = 4us

    

in r24, SREG

push r24                            'Statusregister sichern





   Loadadr Preload , X                  'Adresse von Preload nach X



  $asm



    ld r24, x

    Out Tcnt0 , R24                     'Timer0 mit Preload laden

'   (geht nicht schneller als Timer0=Preload)



    in R24, PortB

    ldi r25,1

    eor r24, r25                        'Bit 0 von r24 invertieren

    Out Portb , R24                     'und am Port ausgeben



    pop r24

    out SREG, r24                              'Statusregister restaurieren

  $end Asm

Return 



�

Timer1 als Timer mit Output Compare

Beispiel: Sekundentakt.



Mit 4MHz - Quarz: TQUARZ  = 1/4(s.



Mit Prescaler = 64 ( 	TCLK = 	TQUARZ ( 64 = 16(s.



Max. Zeit: 	TMAX = 65536 ( 16(s = 1.048576s		(ausreichend).



Nötige Schritte für 1s (Preload - Wert):



	Preload = 1000000(s / 16(s = 62500



Wenn dieser Wert ins Output Compare - Register geschrieben wird, wird nach 1s (Zähler = Preload) ein Interrupt ausgelöst.



Wichtig ist, dass der Timer so konfiguriert wird, dass der Zähler zusätzlich auch rückgesetzt wird:



config timer1 = timer , prescale = 64 , clear timer = 1, compare a = disconnect



Die Option clear timer = 1 bewirkt das Gleiche wie set tccr1b.3, nämlich  Reset Timer1 nach einem Compare Match.



Die Option compare a = disconnect bewirkt, dass am Pin OC1A (=PD5 beim 8535) keine Reaktion bei einem Interrupt stattfindet. Alternativen zu disconnect sind clear / set / toggle.







'Sekundenuhr

‘portD.0 togglet im Sekundentakt, LCD zählt hoch



$crystal = 4000000

$regfile = "8535def.dat"

config pind.0 = output



Config Lcdpin = Pin , Db4 = Portd.4 , Db5 = Portd.5 , Db6 = Portd.6 , Db7 = Portd.7 , Rs = Portd.3 , E = Portd.2

Config Lcd = 20 * 2

Cls



'Timer1 Compare -> jede Sekunde Interrupt

'Reloadwert

dim Newtime as bit

dim secs as long

OCR1AH = High(62500)

OCR1AL = Low(62500)

config timer1 = timer , prescale = 64 , clear timer = 1 , compare a = disconnect

on oc1a timer1_isr

enable oc1a

enable interrupts

'-------------------------------------------------------



Dim W As Word , Channel As Byte



Do

   'Zeit anzeigen

  if Newtime = 1 then

     Newtime = 0



     cls

     locate 1 , 1



     lcd "TIME:" ; secs

  end if





Loop

End



'--------------------------------------------------

Timer1_isr:

    incr SECS

    newtime = 1

    toggle portd.0

return





Timer 2 mit Output Compare

(8bit)



Beispiel: Interrupt alle 2ms



'2ms-Interrupt mit Timer2 Output Compare

'CLK = 4MHz -> TCLK=0.25us

'Prescaler = 64 - > Tclk2 = 64*0.25us = 16us

'OC2 = 2000us/16us = 125



$regfile = "m8def.dat"

$crystal = 4000000



Config Portb = Output



Config Timer2 = Timer , Prescale = 64 , Clear Timer = 1 , Compare = Disconnect

Ocr2 = 125

Enable Oc2

Enable Interrupts

On Oc2 Intcomp2





Do



Loop



End





Intcomp2:

     Toggle Portb.1

Return

�

Timer als Counter



Der Zähleingang liegt beim 8535an

PortB.0 für Timer0 (8bit)

PortB.1 für Timer1 (16bit)



Impulszählung mit Timer1:



'$timer1_ctr1.bas

'zählt Impulse an PortB.1 (Takt-Eingang Timer 1)

'und zeigt den Zählerstand alle 100ms auf dem LCD-Display an



$crystal = 4000000





'Config Timer0 = Counter , Edge = Falling

Config Timer1 = Counter , Edge = Rising



Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



Do

  'Timer0 = 0

  Cls

  Lcd Timer1

  Waitms 100

Loop







Frequenzmessung durch Zählen während 1 Sekunde:



'$timer1_ctr1.bas

'Frequenzmessung an PortB.1 (Takt-Eingang Timer 1)

'und Anzeige alle 100ms auf dem LCD-Display

'max. 65 535Hz durch Zählerauflösung



'Genauigkeit durch Waitms gegeben (nicht exakt)

'Messwerte:

'      100 Hz Anz.  100

'     1000 Hz Anz. 1001

'     10kHz       10010

'



$crystal = 4000000





'Config Timer0 = Counter , Edge = Falling

Config Timer1 = Counter , Edge = Rising



Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



Dim T As Word



Do

  Timer1 = 0

  Waitms 1000

  T = Timer1      'Timer-Inhalt in Variable abspeichern da LCD-Befehle langsam



  Cls

  Lcd T



Loop







Der Timer-Overflow kann benutzt werden, um dies anzuzeigen:



'$timer1_ctr3.bas

'Frequenzmessung an PortB.1 (Takt-Eingang Timer 1)

'und Anzeige alle 100ms auf dem LCD-Display

'max. 65 535Hz durch Zählerauflösung



'Genauigkeit durch Waitms gegeben (nicht exakt)

'Messwerte:

'      100 Hz Anz.  100

'     1000 Hz Anz. 1001

'     10kHz       10010

'



'meldet Overflow



$crystal = 4000000





'Config Timer0 = Counter , Edge = Falling

Config Timer1 = Counter , Edge = Rising

On Timer1 Timer1over

Enable Timer1

Enable Interrupts





Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



Dim T As Word

Dim Overflag As Bit



Do

  Timer1 = 0

  Waitms 1000

  T = Timer1                            'Timer-Inhalt in Variable abspeichern da LCD-Befehle langsam



  Cls

  If Overflag = 0 Then

     Lcd T

     Lcd "Hz"

  Else

     Lcd "Overflow"

     Overflag = 0

  End If



Loop



'-------------------

Timer1over:

     Overflag = 1

     Return



Statt nur einer Overflow-Anzeige  kann auch ein Softwarezähler hochgezählt werden, das erhöht die Auflösung auf 24bit: 



'$timer1_ctr4.bas

'Frequenzmessung an PortB.1 (Takt-Eingang Timer 1)

'und Anzeige alle 100ms auf dem LCD-Display

'max. 65 535Hz durch Zählerauflösung



'Genauigkeit durch Waitms gegeben (nicht exakt)

'Messwerte:

'      100 Hz Anz.  100

'     1000 Hz Anz. 1001

'     10kHz       10010

'    100kHz      100098

'      1MHz     1001448

' bei 10Mhz Fehler durch zu kleinen Abtastrate (2MHz bei 4MHz-Quarz)



'mit zusätzlichem 8bit-Softwarezähler



$crystal = 4000000





'Config Timer0 = Counter , Edge = Falling

Config Timer1 = Counter , Edge = Rising

On Timer1 Timer1over

Enable Timer1

Enable Interrupts





Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2



''Dim T As Word geht nicht, da sonst bei Tg=Tg+T Operanden unterschiedlich

Dim T As Long

Dim Th As Byte

Dim Tg As Long

Dim Overflag As Bit



Do

  	Th = 0

  	Timer1 = 0

 	 Waitms 1000

 	 T = Timer1                                                'Timer-Inhalt in Variable abspeichern da LCD-Befehle langsam



  	Cls

  	Tg = 65536 * Th

  	Tg = Tg + T

    	 Lcd Tg

     	Lcd "Hz"



Loop



'-------------------

Timer1over:

     Th = Th + 1

     Return



Die Genauigkeit ist durch Waitms 1000 begrenzt. Besser ist es, diesen Befehl durch eine exakt abgestimmte ASM-Schleife zu ersetzen.



Die Berechnung solcher Schleifen vereinfacht ein Delay Loop Generator

von Tjabo Kloppenburg (http://www.home.unix-ag.org/tjabo/avr/AVRdelayloop.html).

(Achtung: nach Labels einen Zeilenumbruch einfügen, Kommentare hinter ‘ und nicht ;, sonst meckert BASCOM!)



Erst gilt es aber zu ermitteln, wieviele Taktzyklen an „Eigenverbrauch„ von den 40000000 für 1s abzuziehen sind.



Analyse des generierten Codes mit dem Simulator in AVRStudio:



Th = 0

   +00000032: E080      LDI     R24,0x00          ; R24=0			1 Takt

   +00000033: 93800064  STS     0x0064,R24	     ;[64h]=R24,   Th=0	3 Takte



Timer1=0 	

   +00000035: E080      LDI     R24,0x00          ; R24=0			1 Takt

   +00000036: E090      LDI     R25,0x00          ; R25=0			1 Takt

   +00000037: BD9D      OUT     0x2D,R25	     ; Timer1 LowB = R25 = 0	1 Takt	

   +00000038: BD8C      OUT     0x2C,R24	     ; Timer1 HiB = R24 = 0	1 Takt



Warten 1s

   +00000039: EE88      LDI     R24,0xE8          ; 

   +0000003A: E093      LDI     R25,0x03          ; 3E8h = 1000

   +0000003B: D15E      RCALL   +0x015E           ; Destination: 0x00019A Waitms



T = Timer1

   ;Timer-Adresse laden:

   +0000003C: E4AC      LDI     R26,0x4C          ; 				1 Takt

   +0000003D: E0B0      LDI     R27,0x00          ; X = Timer1-Adr. 2C+20  1 Takt

     ; für RAM-Adress.

   +0000003E: 918D      LD      R24,X+	     ; R24 = [4C], inkr. X	2 Takte

   +0000003F: 919C      LD      R25,X		     ; R25 = [4D]		2 Takte



	;jetzt ist der Timer-Wert abgespeichert: 14 Takte

   

   ;Variablen-Adresse von T laden

   +00000040: E6A0      LDI     R26,0x60          ; 				1 Takt

   +00000041: E0B0      LDI     R27,0x00          ; X = Adresse von T	1 Takt

   +00000042: 938D      ST      X+,R24	     ; R24 -> [60h]		2 Takte

   +00000043: 939D      ST      X+,R25	     ; R25 -> [61h]		2 Takte

   

   ; T ist LONG, also höhere Bytes löschen

   +00000044: 2788      CLR     R24		     ;R24=0			1 Takt

   +00000045: 938D      ST      X+,R24	     ; 0 -> [62h]		2 Takte

   +00000046: 938C      ST      X,R24		     ; 0 -> [63h]		2 Takte





Die Waitms 1000 wird also ersetzt durch:



$asm



' -----------------------------

' warte 3999981 Zyklen:

         ldi  R17, $6B

Wgl0:

         ldi  R18, $46

Wgl1:

         ldi  R19, $B1

Wgl2:

         dec  R19

          brne WGL2

          dec  R18

          brne WGL1

          dec  R17

          brne WGL0

' -----------------------------

' warte 3 Zyklen:

          ldi  R17, $01

Wgl3:

         Dec R17

          brne WGL3

' -----------------------------

' warte 2 Zyklen:

          nop

          nop

' =============================

  $end Asm



�'Messwerte:

Istwert�Anzeige���100 Hz�100 / 99���1000 Hz�1001 / 999���10kHz�9997���100kHz�99973���1MHz�1000193���



Timer als Schwellwert - Auslöser

Wenn der Timer als Counter benutzt wird und mit einem Preloadwert Max - X geladen wird, kann erreicht werden, dass nach einer bestimmten Anzahl von X Impulsen ein Interrupt ausgelöst wird.







Befehle zum Laden: 

	Timer1 =  P

oder  	Load Timer1, X





�Erfassen von Pulslängen

auf die unelegante Art mit Polling:

funktioniert von etwa 5(s....128ms



'$timer1_pulsw1.bas

'Pulsbreite an PortD.2 bestimmen

'

'Genauigkeit ca. +-2us

'Ausgabe seriell RS232



$baud = 2400





Config Pind.2 = Input

Portd.2 = 1                                                 'Pullup ein





Config Timer1 = Timer , Prescale = 8                        'Timer-CLK = 500kHz -> TCLK = 2us

Enable Timer1

On Timer1 Timer1overflow

Enable Interrupts                                           'für Timer1INT





Dim Value As Long

Dim Overflag As Bit



Print "Pulsewidth in us:"



'-------------------

Do

   'Hauptprogramm

      'Warten bis 0 wenn zufällig 1

      Do

      Loop Until Pind.2 = 0





      'Warten auf ansteigende Flanke

      Do

      Loop Until Pind.2 = 1

      Timer1 = 0

      Start Timer1





      'Warten bis abfallende Flanke

      Do

      Loop Until Pind.2 = 0

      Stop Timer1



      'Auswertung

      Value = Timer1

      Value = Value * 2

      If Overflag = 1 Then

          Print "Overflow: >128ms"

      Else

          Print Value;

          Print "us"

      End If

      Overflag = 0



         Waitms 300



Loop

'--------------------

Timer1overflow:

'Bei Timer-Überlauf Flag für Anzeige setzen

Overflag = 1

Return



Ein Herabsetzen des Prescale auf 1 (Auflösung 0.25(s) bringt keine wesentlich höhere Genauigkeit.



Eine Programmierung der Warteschleifen für die Flanken in Assembler bringt höhere Genauigkeit von ca. 1(s.



Programm mit  Funktion Pulsewidth:



'$timer1_pulsw0b.bas

'Pulsbreite an PortD.2 bestimmen

'auf die unelegante Art mit Polling

'mit Prescale=1 und Warten auf Flanke in ASM

'Genauigkeit ca. -1us

'Ausgabe seriell RS232



'Messwerte:

'Sollwert/us   Messwert/us

'   5             4

'  50            49

' 500           499/500

'5000          5000 







$baud = 2400



Declare Function Pulsewidth() As Long



Config Pind.2 = Input

''Portd.2 = 1                             'Pullup ein





Config Timer1 = Timer , Prescale = 1    'Timer-CLK = 4MHz -> TCLK = 0.25us

Enable Timer1

On Timer1 Timer1overflow

Enable Interrupts                       'für Timer1INT





Dim Value As Long

Dim Overflag As Bit



Print "Pulsewidth in us:"



'-------------------

Do

   'Hauptprogramm





      'Auswertung

      Value = Pulsewidth()

      Value = Value / 4

      If Overflag = 1 Then

          Print "Overflow: >128ms"

      Else

          Print Value;

          Print "us"

      End If

      Overflag = 0



         Waitms 300



Loop

'--------------------

Timer1overflow:

'Bei Timer-Überlauf Flag für Anzeige setzen

Overflag = 1

Return



'------------------------------------------------------------------------

Function Pulsewidth() As Long

'Pulsweite bestimmen durch Starten von Timer1 bei ansteigender Flanke und

'Stoppen bei fallender Flanke





'Warten bis 0 wenn zufällig 1

        $asm

Waitlo:

           sbic Pind,2                  'nächsten Befehl überspringen wenn 0

           rjmp WaitLo

       $end Asm





'Warten auf ansteigende Flanke

       $asm

Lohi:

           sbis PIND,2                  'nächsten Befehl überspringen wenn 1

           rjmp LoHi

       $end Asm

      Timer1 = 0

      Start Timer1





'Warten bis abfallende Flanke

       $asm

Hilo:

           sbic Pind,2                  'nächsten Befehl überspringen wenn 0

           rjmp HiLo

       $end Asm

      Stop Timer1

      Pulsewidth = Timer1

End Function

  





Pulsdauermessung über INT0 

Nachteil:	Wegen der Int0-Einsprungzeit muss die Pulsbreite >etwa 50us sein

Vorteil: 	im Hauptprogramm kann der Prozessor etwas anderes tun.



'$timer1_pulsw1.bas

'Messung der Pulsdauer über INT0

'Bei jeder steigenden Flanke an INT0 (PortD.2) wird Timer1 gestartet

'bei jeder fallenden Flanke an INT0 wird Timer1 gestoppt

'die Umschaltung Risig / Falling erfolgt über Edgeflag



'Messwerte:

'Soll       Ist

'    5     Overflow

'   50       50

'  500      500

' 5000     5000

'50000    50016

 

Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portd.7 , E = Portd.6

Config Lcd = 20 * 2





Config Pind.2 = Input



On Int0 Int0serv                        'Nosave geht nicht!

Config Int0 = Rising

Enable Int0



Config Timer1 = Timer , Prescale = 8    'Timer-CLK = 500kHz -> TCLK = 2us

Enable Timer1

On Timer1 Timer1overflow



Dim Value As Long

Dim Overflag As Bit

Overflag = 0

Dim Edgeflag As Bit

Edgeflag = 0



Cls

Lcd "Measure pulsewidth"





'-------------------

Do

   'Hauptprogramm

      Disable Interrupts

      Value = Value * 2

      Locate 2 , 1

      If Overflag = 1 Then

         Lcd "Overflow: >128ms"

      Else

         Lcd Value;

         Lcd "us           "

      End If

      Overflag = 0

      Enable Interrupts

      Waitms 300

Loop

'--------------------



Int0serv:

 If Edgeflag = 0 Then

   'steigende Flanke

    Edgeflag = 1                        'bereit für abfallende

    Config Int0 = Falling

    Timer1 = 0

    Start Timer1

  Else

    'fallende Flanke

    Edgeflag = 0                        'bereit für steigende Flanke

    Config Int0 = Rising

    Stop Timer1

    Value = Timer1                      'in 2 us

  End If

Return

‘----------------------

Timer1overflow:

   Overflag = 1

Return







Periodendauermessung

Prinzip:

Warten bis Eingang = 0: definierter Beginn�Unbedingt notwendig da sonst Fehler!��Warten bis Eingang = 1�1. ansteigende Flanke��Zähler starten���Warten bis Eingang = 0  �abfallende Flanke��Warten bis Eingang = 1�2. ansteigende Flanke��Zähler stop���Zähler auslesen und Wert ggf. umrechnen���

'$timer1_period1.bas



'Periodendauer an PortD.2 bestimmen

'auf die unelegante Art mit Polling

'mit Prescale=1 und Warten auf Flanken in ASM (Funktion Period())

'Genauigkeit ca. -1us

'Ausgabe seriell RS232



'Messungen:

'Sollwert/us   Messwert/us

'          10           9

'         100          99

'        1000         999 / 1000

'       10000       10003 / 1000

'      100000      100040

'     1000000     1000409       Abweichung möglicherweise durch Ungenauigkeit der Referenz



'Riesenfehler!!!!

'funktioniert aber im Simulator richtig!





$baud = 2400

$crystal = 8000000





Declare sub MeasurePeriod()



Config Pind.2 = Input

''Portd.2 = 1                             'Pullup ein





Config Timer1 = Timer , Prescale = 1                        'Timer-CLK = 8MHz -> TCLK = 0.125us

Enable Timer1

On Timer1 Timer1overflow

Enable Interrupts                                           'für Timer1INT



'Variablen für Periodendauer u.Frequenz:

Dim ValueL As Word at &h60

dim ValueH as word at &h62

dim T as long at &h60 overlay

dim tsingle as single

dim f as single



Print "Period in us:"



'-------------------

Do

   'Hauptprogramm





      'Auswertung

      MeasurePeriod

      tsingle = t

      tsingle = tsingle / 8

      T = T \ 8

      f = 1000000 / T

      print T;

      print "    T=";

      print tsingle;

      print "us        f=";

      print f;

      print "Hz"



         Waitms 500



Loop



'------------------------------------------------------------------------

sub MeasurePeriod()

'Periodendauer bestimmen durch Starten von Timer1 bei ansteigender Flanke und

'Stoppen bei der nächsten



'schreibt Ergebnis in ValueH und ValueL

'benutzt Timer1 und Timer1-Interrupt



ValueH = 0



'Wenn zufällig schon 1:warten bis null (wichtig, sonst komische Fehler!)

       $asm

Waitforlow:

           sbic Pind,2

           rjmp Waitforlow

       $end Asm



'Warten auf 1. ansteigende Flanke (Schleife solange 0)

      $asm

Lohi1:

           sbis PIND,2                                      'nächsten Befehl überspringen wenn 1

           rjmp LoHi1

       $end Asm

      Timer1 = 0

      Start Timer1





'Warten bis abfallende Flanke

       $asm

Hilo:

           sbic Pind,2                                      'nächsten Befehl überspringen wenn 0

           rjmp HiLo

       $end Asm





'Warten auf 2. ansteigende Flanke

       $asm

Lohi2:

           sbis PIND,2                                      'nächsten Befehl überspringen wenn 1

           rjmp LoHi2

       $end Asm



      Stop Timer1

      ValueL = Timer1

End sub

'--------------------

Timer1overflow:

incr ValueH

Return



Wenn nur Timer0 zur Verfügung steht, muss normalerweise die Zählerbreite softwaremässig erhöht werden:

Testprogramm zur Periodendauermessung über INT0 und Timer0:



'Testprogramm für T-Messung über Timer0/Int0

'Messwerte:

'Tsoll     Tmess   /us

'10        keine Reaktion (Int nicht schnell genug)

'100       98

'1000      998/999

'10000     10010 

'100000    100041

'1000000   1000424





'---------------------------------------------------

$regfile = "2313def.dat"

$crystal = 8000000

$baud = 2400





'Periodendauermessung benutzt Int0 und Timer0

Config Pind.2 = Input

On Int0 Int0serv

Config Int0 = Rising

Enable Int0



'Timer0 für Periodendauermessung

Config Timer0 = Timer , Prescale = 8                        'Timer-CLK = 8MHz -> TCLK = 1us

Enable Timer0                                               'Overflow alle  256us

On Timer0 Timer0overflow



Enable Interrupts

'-----------------------------------------------------



'Variablen Periodendauermessung

Dim T As Long At &H60                                       'Periodendauer

Dim Tvaluel As byte At &H60 Overlay                         'Periodendauer low byte

Dim Tvalueh As word At &H61 Overlay                         'Periodendauer high word

Dim T0overflowctr As Word                                   'Zähler für Timer1Overflow



T = 0



'--------------------------------------------------------------------

Do



      Print "T=";

      print T;

      print "us"



     waitms 1000



Loop

'---------------------------------------------------------------------



Int0serv:                                                   'Impulsflanke Lichtschranke

'Periodendauermessung

 'Werte der alten Periodendauermessung übernehmen:

    Stop Timer0

    Tvaluel = Timer0                                        'Low word von T

    Tvalueh = T0overflowctr                                 'High word von T



 'neue Periode messen:

     Timer0 = 0

     Start Timer0

     T0overflowctr = 0



Return



'-----------------------------------



Timer0overflow:

   Incr T0Overflowctr                                       'wieder auf 1.Flanke warten

return

'-------------------------------------------



 



PWM über Timer-Overflow



'$PWM_Timer0.BAS

'PWM über Timer0, fester Wert in %: 0...100



Config Portb = Output



'Timer0 für PWM

'Timer-CLK = 4MHz , Overflowfreq ca. 16kHz , '->PWM-Freq. bei 256 Schritten ca. 61Hz

'            						bei 100 Schritten 160Hz

Config Timer0 = Timer , Prescale = 1

Enable Timer0

On Timer0 Timer0overflow

Enable Interrupts



Dim Loopctr As Byte

Dim Pwm As Byte



Pwm = 20

'-------------------

Do

Loop

'--------------------

Timer0overflow:

   If Pwm > Loopctr Then

       Portb.0 = 1

   Else

       Portb.0 = 0

   End If

   Loopctr = Loopctr + 1



   'Begrenzung auf 100 Schritte:

   If Loopctr = 100 Then Loopctr = 0



Return

�PWM mit  Compare-Funktion des Timers

Testprogramm mit 20/255 Tastgrad:



'$timer1_pwm1.bas

'PWM-Signal mit 20/255 an OC1 (=PB3 bei 2313, =PD5 beim 8535)

'Config Pind.5 = Output (nicht nötig)

'Timer1 als PWM mit

'  Prescale=8 -> fCTR = FXTAL/8 = 500kHz

'                mit 2^8-2 = 510 Schritten -> 980Hz PWM-Frequenz 

'  PWM = 8  -> 8bit PWM    (möglich: 8/9/10 bit)

'  Compare A Pwm = Clear Down  -> Ausgang low beim Herunterzählen (nichtinvertiertes PWM)

Config Timer1 = Pwm , Prescale = 8 , Pwm = 8 , Compare A Pwm = Clear Down



'Wert

Pwm1a = 20



Do

Loop



PWM-Steuerung über die serielle Schnittstelle mit Eingabe des Wertes in %:

'$timer1_pwm2.bas

'PWM-Signal mit Steuerung über RS232



$crystal = 4000000

$baud = 2400



Dim Value As Byte

Dim Temp As Word





'Timer1 als PWM mit

'  Prescale=8 -> fCTR = FXTAL/8 = 500kHz

'                 mit 2^(8+1)-2 = 510 Schritten -> 980Hz PWM-Frequenz

'  PWM = 8  -> 8bit PWM    (möglich: 8/9/10 bit)

'  Compare A Pwm = Clear Down  -> Ausgang low beim Herunterzählen (nichtinvertiertes PWM)

Config Timer1 = Pwm , Prescale = 64 , Pwm = 8 , Compare A Pwm = Clear Down





'Wert

Value = 0

Temp = Value * 255

Temp = Temp / 100

Pwm1a = Temp





Do

  Input "Value (0...100)" , Value

  Print Value

  Temp = Value * 255

  Temp = Temp / 100

  Pwm1a = Temp

  Print Temp

Loop





Testprogramm für PWM mit Timer2



'PWM mit Timer2, Prinzip (fester Wert)

'Ausgang an OC2(=PD7 beim 8535)

$regfile = "8535def.dat"

$crystal = 4000000



'config portb = output



config timer2 = PWM , PRESCALE = 8 , PWM = ON , COMPARE PWM = CLEAR UP

OCr2 = 100                    'Fest auf 100/256

enable timer2



do



loop

end



PWM mit Timer2 und Up/Down über Taster



$regfile = "8535def.dat"

$crystal = 4000000



config portb = input

portb.0 = 1                   'Pullup

portb.1 = 1                   'Pullup





config timer2 = PWM , PRESCALE = 1 , PWM = ON , COMPARE PWM = CLEAR UP

OCr2 = 100                    'Anfangswert auf 100/256

enable timer2



do

       debounce Pinb.0 , 0 , Increment , sub

       debounce Pinb.1 , 0 , Decrement , sub

loop



sub increment

  if ocr2 < 246 then

      ocr2 = ocr2 + 10

  end if

end sub



sub decrement

  if ocr2 > 9 then

      ocr2 = ocr2 - 10

  end if

end sub





Erzeugung von zwei unabhängigen PWM-Signalen mit Timer1



$regfile = "m8def.dat"

$crystal = 4000000





Config Timer1 = Pwm , Prescale = 1 , Pwm = 8 , Compare A Pwm = Clear Down , Compare B Pwm = Clear Up

Pwm1a = 20

Pwm1b = 30





Do

Loop



Ausgang OC1A: 20/256 active high

Ausgang OC1B: 30/256  active low



�Timer-Register (Mega8)



Gemeinsame Register für alle Timer:



TIMSK	Timer/Counter Interrupt Mask Register 

schaltet für die 3 Timer gemeinsam die Timer-Interrupts ein und aus (1 = Interrupt eingeschaltet)



Bits     7	6	  5	    4	        3	          2	1	   0

OCIE2�TOIE2�TCIE1�OCIE1A�OCIE1B�TOIE1�-�TOIE0��

Timer0:

TOIE = Timer0 Overflow Output Enable   



Timer1:

TICIE1	= Timer/Counter1 Input Capture Interrupt Enable

OCIE1A 	= Timer/Counter1 Output Compare A Match Interrupt Enable

OCIE1B	= Timer/Counter1 Output Compare B Match Interrupt Enable

TOIE1 	= Timer/Counter1 Overflow Interrupt Enable



Timer2:

OCIE2		= Timer/Counter2 Output Compare Match Interrupt Enable

TOIE2 	= Timer/Counter2 Overflow Interrupt Enable



TIFR	Timer/Counter Interrupt Flag Register 



Bits     7	6	  5	    4	        3	          2	1	   0

OCF2�TOV2�ICF1�OCF1A�OCF1B�TOV1�-�TOV0��

Timer0:

TOV0 = Timer0 Overflow Flag   



Timer1:

TOV1		= Timer/Counter1 Overflow Flag

OCF1A 	= Timer/Counter1 Output Compare A Match Flag

OCF1B	= Timer/Counter1 Output Compare B Match Flag

ICF1	 	= Timer/Counter1 Input Capture Flag



Timer2:

TOV2		= Timer/Counter2 Overflow Flag

OCF2	 	= Timer/Counter2 Output Compare Flag





Timer0



TCCR0	 Timer/Counter 0 Control Register



legt den Prescaler-Faktor fest, und schaltet den Timer ein und aus



	Bits     7	6	  5	    4	        3	          2	1	   0

�����CS02�CS01�CS00��

CS02�CS01�CS00�Prescaler��0�0�0�Timer gestoppt��0�0�1�1��0�1�0�8��0�1�1�64��1�0�0�256��1�0�1�1024��1�1�0�Ext. Takt an T0, fallende Flanke��1�1�1�Ext. Takt an T0, ansteigende Flanke��



TCNT0	 Timer/Counter Register

Zählerregister 8 bit



Bits     7	6	  5	    4	        3	          2	1	   0			 

���������





Timer1



TCCR1	 Timer/Counter 1 Control Register 



TCCR1A	 Timer/Counter 1 Control Register A:



Bits     7	6	  5	    4	        3	          2	1	   0

COM1A1�COM1A0�COM1B1�COM1B0�FOC1A�FOC1B�WGM11�WGM10��

TCCR1B	 Timer/Counter 1 Control Register B:



Bits     7	6	  5	    4	        3	          2	1	   0

ICNC1�ICES1�-�WGM13�WGM12�CS12�CS11�CS10��

COM1: Compare Output Mode for channel A/B

stellen das  Verhalten der Pins OC1A, OC1B ein.

Eine 1 bewirkt, dass das normale Verhalten des Pins aufgehoben wird und ein Ausgangssignal erzeugt wird, wenn der Port als Ausgang geschaltet ist (Bit im DDR gesetzt).



 	Verhalten der OC-Pins (ausser PWM-Modus):

COM1x1�COM1x0���0�0�OC1-Pins funktionieren normal (disconnected)��0�1�Toggle on compare match��1�0�Clear on compare match��1�1�Set on compare match��		mit x = A/B

Verhalten der OC-Pins im Fast PWM-Modus:

COM1x1�COM1x0���0�0�OC1-Pin funktioniert normal (disconnected)��0�1�Wenn WGM13:0 = 15:

Toggle OC1A bei Compare Match,�OC1B disconnected�

ansonsten:�OC1-Pins funktionieren normal (disconnected)��1�0�Clear on compare match�Set at TOP��1�1�Set on compare match�Clear at TOP��		mit x = A/B



Verhalten der OC-Pins im normalen PWM-Modus:

COM1x1�COM1x0���0�0�OC1-Pin funktioniert normal (disconnected)��0�1�Wenn WGM13:0 = 9 oder 14:

Toggle OC1A bei Compare Match,�OC1B disconnected�

ansonsten:�OC1-Pins funktionieren normal (disconnected)��1�0�Clear on compare match beim Aufwärtszählen�Set on compare match beim Abwärtszählen��1�1�Set on compare match beim Aufwärtszählen�Clear on compare match beim Abwärtszählen��		mit x = A/B



FOC1: Force Output Compare for channel A/B

(nur schreiben, gelesen wird immer 0)



WGM13....WGM10	Waveform Generation Mode

stellen PWM-Modus ein



	0000 = PWM aus

	0001 = PWM 8 bit

	0010 = PWM 9 bit 

	0011 = PWM 10 bit 

	weitere siehe Datenblatt



ICNC1: Input Capture Noise Canceller

	Filterung des ICP1-Pins



ICES1: Input Capture Edge Select

	Triggermodus für Input Capture

	0 = fallende Flanke

	1 = ansteigende Flanke 



CS12, CS11, CS10: Clock Select

	wie bei Timer0





�

TCNT1	 Timer/Counter Register

Zählerregister 16 bit



Bits     7	6	  5	    4	        3	          2	1	   0			 

��������TCNT1H����������TCNT1L��



OCR1A, OCR1B	 Output Compare Register A und B 

Vergleichsregister 16 bit,wird mit TCNT1 verglichen. 

Bei Übereinstimmung kann ein Interrupt ausgelöst, oder der Zustand der OC1-Pins verändert werden.



Bits     7	6	  5	    4	        3	          2	1	   0			 

��������OCR1H����������OCR1L��



ICR1	 Input Capture Register 

Wenn ein Capture-Ereignis am ICP1-Pin auftritt (oder optional am Komparator-Ausgang), wird der Inhalt des Timer-Registers TCNT1 ins Captureregister ICR1  geschrieben (16 bit)



Bits     7	6	  5	    4	        3	          2	1	   0			 

��������OCR1H����������OCR1L��





Timer2



TCCR2	 Timer/Counter 2 Control Register 



Bits     7	6	  5	    4	        3	          2	1	   0

FOC2�WGM20�COM21�COM20�WGM21�CS22�CS21�CS20��



FOC2: Force Output Compare 

(nur schreiben, gelesen wird immer 0)



WGM21, WGM20	Waveform Generation Mode

stellen PWM-Modus ein

	

WGM21�WGM20���0�0�Normal��0�1�PWM phasenkorrekt��1�0�Clear Timer on Compare Match��1�1�Fast PWM��





COM21, COM20: Compare Output Mode for channel A/B

stellen das  Verhalten des Pins OC2 ein.

Eine 1 bewirkt, dass das normale Verhalten des Pins aufgehoben wird und ein Ausgangssignal erzeugt wird, wenn der Port als Ausgang geschaltet ist (Bit im DDR gesetzt).



 	Verhalten der OC-Pins (ausser PWM-Modus):

COM21�COM20���0�0�OC2-Pin funktioniert normal (disconnected)��0�1�Toggle OC2 on compare match��1�0�Clear OC2 on compare match��1�1�Set OC2 on compare match��		



	Verhalten der OC-Pins im normalen (phasenkorrekten) PWM-Modus:

COM21�COM20���0�0�OC2-Pin funktioniert normal (disconnected)��0�1�-��1�0�Clear on compare match�Set at TOP��1�1�Set on compare match�Clear at TOP��		



Verhalten der OC-Pins im normalen PWM-Modus:

COM21�COM20���0�0�OC2-Pin funktioniert normal (disconnected)��0�1�-��1�0�Clear on compare match when upcounting�Set on compare match when downcounting��1�1�Set on compare match when upcounting�Clear on compare match when downcounting��		





CS22, CS21, CS20: Clock Select

Achtung: ähnlich wie bei Timer0, aber z.T. andere Teilverhältnisse!



CS22�CS21�CS20�Prescaler��0�0�0�Timer gestoppt��0�0�1�1��0�1�0�8��0�1�1�32��1�0�0�64��1�0�1�128��1�1�0�256��1�1�1�1024��



TCNT2	 Timer/Counter Register

Zählerregister 8 bit



Bits     7	6	  5	    4	        3	          2	1	   0			 

����������



OCR2	 	Output Compare Register 

Vergleichsregister 8 bit,wird mit TCNT2 verglichen. 

Bei Übereinstimmung kann ein Interrupt ausgelöst, oder der Zustand der OC2-Pins verändert werden.



Bits     7	6	  5	    4	        3	          2	1	   0			 

����������



ASSR	 	Asynchronous Status Register 

erlaubt den Betrieb mit internem Takt 1MHz für den Prozessor und Uhrenquarz 32768Hz für Timer2.  



Bits     7	6	  5	    4	        3	          2	1	   0			 

-�-�-�-�AS2�TCN2UB�OCR2UB�TCR2UB���

AS2 = 	1: asynchrone Operation mitb Uhrenquarz

weitere Détails siehe Datenblatt.	�

Analogkomparator

für den 2313:

�

ACSR = Analog Comparator Status Register



ACSR.7 = 0 ( Komparator ein  (beim Umschalten ACIE = 0 damit kein Interrupt ausgelöst wird)

ACSR.6 = 1 (  interne Referenz 1.23V an beim Mega8

ACSR.5 = Komparator-Output

ACSR.3 = 1 ( Komparator-Interrupt enabled

ACSR.2 = 1 ( Komparator Input Capture enabled

ACSR.1,0  ( Interrupt Mode



�



Beispiel:

$regfile = "2313def.dat"

$crystal = 4000000

Config Portd = Output



Acsr = 0                                          'Komparator ein, vergleicht AIN1 und AIN0



Do

Portd.0 = Acsr.aco

Loop



Achtung: Bug! In M8def.dat war ACSR.ACO nicht definiert (3.1.2007)

�

Gleiches Beispiel mit Interrupt:

'Komparator reagiert bei Änderung an PinB.3 (AIN1) und überträgt B.3 auf A.7

$hwstack = 40

$crystal = 4000000

$regfile = "8535def.dat"

$baud = 9600



Config Pina.7 = Output



'Komparator Interrupt enabled, Interrupt by pos. u. neg. Flanke

Acsr = &B00001000

Enable Interrupts

On Aci Oncomparator

'----------------------------

Do

Loop

'----------------------------

Oncomparator:

Porta.7 = Acsr.aco

Return





Analog-Digital-Wandler

Prinzip: nach dem Wägeverfahren

Auflösung: 10bit

Linearität: 0.5bit, Genauigkait ( 2 bit

Dauer: 65...260(s beim ATMega8, abhängig von der Auflösung.

Volle Auflösung bei 50...200kHz ADC-Takt

Eine Konversion dauert 13  AD-Wandler-Taktzyklen, ausser beim ersten Mal, dann sind es 25.



Achtung: 

ADC4, ADC5 beim AtMega8 haben nur 8 bit Auflösung!

AVCC darf sich um max. 0.3V von VCC unterscheiden (Schottky-Diode intern)�Am besten AVCC aus VCC über Tiefpass 10(H / 100nF

AREF mit 100nF abblocken

Signalquelle sollte <10k( Impedanz haben



Berechnung des Spannungswertes:

�



ADMUX-Register (M8):

�



�

Achtung: interne Referenz nicht benutzen wenn an AREF eine Spannung anliegt!



ADLAR: rechts- oder linksbündige Darstellung im Register

ADLAR = 0:

�

ADLAR = 1:

�



MUX:

�



ADCSRA:

�



ADEN = 1: ADC Enable

ADSC = 1: Start Conversion

ADFR = 1: Free running mode selected

ADIF: AD-Wandler-Interrupt flag. Wird gesetzt wenn die Wandlung abgeschlossen ist.�	Löst den Interrupt aus wenn ADIE = 1. (Flag wird automatisch rückgesetzt)

ADIE = 1: AD Interrupt enable

ADP: Prescaler 

�





Beispiel 1:

'Kanal 1 nur 8 bit lesen

'ADC freilaufend



$regfile = "m8def.dat"

$crystal = 3686400

$baud = 9600





Config Portc = Input

Admux = &B01100001                              '01 externe Referenz, 1 nur ADCH 8 bit, MUX = 1

Adcsra = &B11100010                             '1 ADEN, 1 Start, 1 Free run,

                                                 	 '00 IntFlag = 0, kein Interrupt

                                                 	 '010 Prescaler = 4

Do

   Print Adch

   Waitms 2

Loop



Beispiel 2

'ADC freilaufend

'mit Interrupt

'schnelle Übertragung und langsamer Takt damit keine Messwerte verloren gehen



$regfile = "m8def.dat"

$crystal = 3686400

$baud = 38400



On Adc Adcint



Config Portc = Input

Admux = &B01100001             '01 externe Referenz, 1 nur ADCH 8 bit, MUX = 1



Adcsra = &B11101111              '1 ADEN, 1 Start, 1 Free run,

                                                  '0 IntFlag = 0, 1 Interrupt

                                                  '111 Prescaler = 128



Enable Interrupts

Do



Loop



Adcint:

   Print Adch

   Return





Beispiel 3:

'Kanal 1 nur 8 bit lesen

'ADC im Single-Betrieb

'mit Interrupt





$regfile = "m8def.dat"

$crystal = 3686400

$baud = 38400



On Adc Adcint



Config Portc = Input

Admux = &B01100001                                '01 externe Referenz, 1 nur ADCH 8 bit, MUX = 1



Adcsra = &B10001111                               '1 ADEN, 0 kein Start, 0 Single,

                                                  '0 IntFlag = 0, 1 Interrupt

                                                  '111 Prescaler = 128



Enable Interrupts

Do

     Set Adcsra.adsc                              'start conversion

     Waitms 100

Loop



Adcint:

   Print Adch

   Return





'Kanal 1 nur 8 bit lesen

'ADC SINGLE Betrieb ohne Interrupt



$regfile = "m8def.dat"

$crystal = 3686400

$baud = 9600





Config Portc = Input

Admux = &B01100001                                '01 externe Referenz, 1 nur ADCH 8 bit, MUX = 1

Adcsra = &B11000010                               '1 ADEN, 1 Start, 0 kein Free run,

                                                  '00 IntFlag = 0, kein Interrupt

                                                  '010 Prescaler = 4

Do

   Set Adcsra.adsc

   'Warten bis Wandlung fertig

   Do

   Loop Until Adcsra.adsc = 0



   Print Adch

   Waitms 100

Loop







'AD-Wandler 1 Kanal

'AREF = +5V

'Achtung: AGND an Masse AVCC = +5V



$crystal = 8000000

$regfile = "8535def.dat"



Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portc.3 , E = Portc.2

Config Lcd = 20 * 2

Cls



'_____________________________________________________



'ADC konfigurieren: Auto ergibt höchste Taktrate, Single muss sein für GetADC

Config Adc = Single , Prescaler = Auto

'ADC starten

Start Adc



Dim W As Word , Channel As Byte



Channel = 0



Do

  W = Getadc(channel)

  cls

  LCD " value " ; W

  waitms 100



Loop

End



�

'JEDE SEKUNDE MESSWERT ANZEIGEN

'AD-Wandler 1 Kanal als Voltmeter 0...5V

'AREF = +5V

'Achtung: AGND an Masse AVCC = +5V





$crystal = 4000000

$regfile = "8535def.dat"



config Pind.0 = output



'Timer1 als Sekundentakt

OCR1AH = high(62500)

OCR1AL = Low(62500)

config timer1 = timer , prescale = 64 , clear timer = 1 , compare a = disconnect

On OC1A CountSeconds

ENABLE INTERRUPTS

ENABLE OC1A

dim CSecs as long

dim Newtime as bit



'LCD

Config Lcdpin = Pin , Db4 = Portc.4 , Db5 = Portc.5 , Db6 = Portc.6 , Db7 = Portc.7 , Rs = Portc.3 , E = Portc.2

Config Lcd = 20 * 2

Cls



'ADC konfigurieren: Auto ergibt höchste Taktrate, Single muss sein für GetADC

Config Adc = Single , Prescaler = Auto

'ADC starten

Start Adc

Dim W As Word , Channel As Byte

dim U as single

dim Us as string * 10



Channel = 0



Do



   'Zeit anzeigen

  if Newtime = 1 then

     Newtime = 0



     'Spannung messen

     W = Getadc(channel)

     u = w * 5

     u = u / 1024





     'und anzeigen:

     cls

     locate 1 , 1



     lcd "TIME:" ; Csecs

     

     locate 2 , 1

     Us = fusing(u , "#.###")

     LCD "U=" ; Us



  end if





Loop

End



'--------------------------------------------------

CountSeconds:

    toggle portd.0

    incr cSECS

    newtime = 1

return

�Variablen, Stack und Frame



�0FFF�����XRAM  (extern fakultativ)���RAMEND+1���Hardware-Stack�RAMEND���Software-Stack����Frame Space��SRAM������Variablen�����60h����5Fh���max. 64 I/O-Register��I/O���20h����1Fh���32 Prozessor-Register��Register���0���



BASCOM benutzt das SRAM  des Kontrollers für folgende Zwecke:�

zum Speichern von Variablen ab Adresse 60h d.h. oberhalb von Registerbereich (0...1Fh) und IO-Bereich (20h...5Fh).



Speicherverbrauch:



Variablentyp�Speicherverbrauch�Wertebereich��8 Bit�1 Byte���Byte�1 Byte�0...255��Integer�2 Byte�-32768...+32767��Word�2 Byte�0...65535��Long�4 Byte�-2 147 483 648...2 147 483 647�� Single�4 Byte�(1.5 x 10–45 ...3.4 x 1038 ��String der Länge L�L+1 Byte�max. 254 Zeichen��

�

als Hardware-Stack zum Speichern der Rücksprungadresse bei CALL und von Registerinhalten die mit PUSH gerettet wurden.�Der Hardwarestack wächst vom oberen SRAM-Ende (DFh für den 2313) nach unten.



1 CALL braucht 2 Byte

1 PUSH braucht 1 Byte��Das heisst der Aufruf einer Interrupt-Routine ohne NOSAVE braucht 2+29 Byte Stack, da  SREG , und R0 bis R31 ausser R12 bis R15 gesichert werden.



�

als Software-Stack zum 

Speichern der Adressen von LOCAL-Variablen.�Jede LOCAL-Variable benötigt 2 Byte zum Speichern der Adresse.�

Übergeben von Variablen an einen SUB oder eine FUNCTION



SOFT-STACK SIZE = (n+m)*2 + 4

n = max. Anzahl der Parameter von SUBs/FUNCTIONs

m = max. Anzahl von LOCAL Variablen

Zur Sicherheit wird 4 hinzuaddiert. �

als Frame 

zum Speichern von LOCAL-Variablen.�z.B. braucht ein String von 40 Zeichen 40+1(Stringende)+4 (Sicherheit) = 45 Byte.�

für Umrechnungsroutinen wie STR(), VAL�16 Byte�

für INPUT, PRINT, LCD�wenn diese mit numerischen Variablen gebraucht werden, da dann eine Konversion stattfindet. 

 



Die Default-Einstellung ist  32 Byte Hardware-Stack, 8 Byte Soft Stack, 16 Byte Frame:

für den ‘2313:



�00�01�02�03�04�05�06�07�08�09�0A�0B�0C�0D�0E�0F��60������������������70������������������80������������������90������������������A0���������FRAME��B0�FRAME�SOFT   STACK��C0�HARDWARE��D0�STACK��



Wenn Interrupts benutzt werden, sollte der Hardware-Stack vergrössert werden, da ein Interrupt-Aufruf schon 31 Byte braucht!







Externes RAM

Die XRAM-Adressen beginnen bei 260h, da unterhalb die internen Adressen liegen. Es wird also ein Teil des externen RAMs nicht genutzt.



�

Hard- und Software-Stack wachsen nach unten, von RAMEND aus gesehen.

RAMEND ist abhängig vom Kontroller:



�AT90S1200�AT90S2313�AT90S4414�AT90S8515�AT90S8535���RAM-Grösse�nur Stack�128�256�512�512���RAMEND�5F�DF�15F�25F����

